4 - Problemi e modelli su grafi

Mauro Passacantando

Dipartimento di Scienze Economico-Aziendali e Diritto per l'Economia
Università degli Studi di Milano-Bicocca
mauro.passacantando@unimib.it

Corso di Dinamica dei Sistemi Aziendali
Laurea Magistrale in Scienze Economico-Aziendali
Università degli Studi di Milano-Bicocca

Grafi orientati

Un grafo orientato è una coppia di insiemi (N, A), dove A è un insieme di coppie ordinate di elementi di N.
N è chiamato l'insieme dei nodi, mentre A è l'insieme degli archi.

Per ogni arco (i, j), il nodo i è detto la coda dell'arco, mente il nodo j è la testa.

Esempio. Il grafo $G = (N, A)$, dove $N = \{1, 2, 3, 4\}$ e $A = \{(1, 2), (1, 3), (2, 4), (3, 4)\}$ può essere rappresentato così:



Assegnamento di operai ad attività - descrizione

Il dirigente di un'azienda ha 10 attività da svolgere la prossima settimana e 8 operai a disposizione per svolgerle. Gli operai hanno competenze diverse, per cui ognuno di loro è in grado di svolgere solo alcune delle 10 attività, come indicato in tabella:

	Attività									
Operai	1	2	3	4	5	6	7	8	9	10
1	sì	no	no	sì	sì	no	no	sì	no	sì
2	no	sì	no	no	sì	sì	no	no	sì	no
3	sì	no	sì	no	sì	no	sì	sì	no	sì
4	no	no	no	sì	sì	no	no	sì	no	no
5	sì	no	no	sì	no	sì	no	sì	no	sì
6	sì	sì	no	sì	sì	no	sì	no	sì	sì
7	no	no	sì	sì	sì	no	no	sì	no	sì
8	sì	no	sì	no	no	sì	no	no	no	no

Sapendo che ogni operaio può svolgere solo un'attività, e che ogni attività può essere svolta solo da un operaio, il dirigente deve assegnare gli operai alle attività in modo da eseguire il massimo numero possibile di attività.

Assegnamento di operai ad attività - formulazione

Se O è l'insieme degli operai e T l'insieme delle attività, allora possiamo definire un grafo bipartito (N, A), dove
l'insieme dei nodi $N = O \cup T$,
l'insieme degli archi $A = \{(i,j) : \text{ operaio } i \in O \text{ è in grado di svolgere l'attività } j \in T\}$



Assegnamento di operai ad attività - formulazione

Variabili decisionali: $x_{ij} = \begin{cases} 1 & \text{se assegno l'operaio } i \text{ all'attività } j \\ 0 & \text{altrimenti} \end{cases}$

per ogni arco $(i,j) \in A$.

Funzione obiettivo: $\sum_{(i,j)\in A} x_{ij}$ (da massimizzare)

Vincoli:

ogni operaio può svolgere solo un'attività:

$x_{1,1} + x_{1,4} + x_{1,5} + x_{1,8} + x_{1,10} \leq 1$

$x_{2,2} + x_{2,5} + x_{2,6} + x_{2,9} \leq 1$

$x_{3,1} + x_{3,3} + x_{3,5} + x_{3,7} + x_{3,8} + x_{3,10} \leq 1$

$x_{4,4} + x_{4,5} + x_{4,8} \leq 1$

$x_{5,1} + x_{5,4} + x_{5,6} + x_{5,8} + x_{5,10} \leq 1$

$x_{6,1} + x_{6,2} + x_{6,4} + x_{6,5} + x_{6,7} + x_{6,9} + x_{6,10} \leq 1$

$x_{7,3} + x_{7,4} + x_{7,5} + x_{7,8} + x_{7,10} \leq 1$

$x_{8,1} + x_{8,3} + x_{8,6} \leq 1$

Assegnamento di operai ad attività - formulazione

Vincoli:

ogni attività può essere svolta solo da un operaio:

$x_{1,1} + x_{3,1} + x_{5,1} + x_{6,1} + x_{8,1} \leq 1$

$x_{2,2} + x_{6,2} \leq 1$

$x_{3,3} + x_{7,3} + x_{8,3} \leq 1$

$x_{1,4} + x_{4,4} + x_{5,4} + x_{6,4} + x_{7,4} \leq 1$

$x_{1,5} + x_{2,5} + x_{3,5} + x_{4,5} + x_{6,5} + x_{7,5} \leq 1$

$x_{2,6} + x_{5,6} + x_{8,6} \leq 1$

$x_{3,7} + x_{6,7} \leq 1$

$x_{1,8} + x_{3,8} + x_{4,8} + x_{5,8} + x_{7,8} \leq 1$

$x_{2,9} + x_{6,9} \leq 1$

$x_{1,10} + x_{3,10} + x_{5,10} + x_{6,10} + x_{7,10} \leq 1$

Assegnamento di operai ad attività - formulazione

Modello matematico (matching di massima cardinalità)

$$
\begin{cases}
\max \displaystyle\sum_{(i,j)\in A} x_{ij} \\
\displaystyle\sum_{j\in T:\ (i,j)\in A} x_{ij} \leq 1 \qquad \forall\ i \in O \\
\displaystyle\sum_{i\in O:\ (i,j)\in A} x_{ij} \leq 1 \qquad \forall\ j \in T \\
x_{ij} \in \{0,1\} \qquad \forall\ (i,j) \in A
\end{cases}
$$

Assegnamento di operai ad attività - formulazione AMPL

assegn1.mod

```
#----------- parametri --------------------

set O; # insieme degli operai
set T;  # insieme delle attività
set A within (O cross T); # insieme degli archi

#----------- variabili --------------------

var x{(i,j) in A} binary;

#----------- funzione obiettivo --------------------

maximize AttivitaSvolte: sum{(i,j) in A} x[i,j];

#----------- vincoli ------------------

s.t. v_operai{i in O}: sum{j in T: (i,j) in A} x[i,j] <= 1;

s.t. v_attivita{j in T}: sum{i in O: (i,j) in A} x[i,j] <= 1;
```

Assegnamento di operai ad attività - formulazione AMPL

assegn1.dat

```
set O := 1 2 3 4 5 6 7 8;

set T := 1 2 3 4 5 6 7 8 9 10;

set A := (1,1)  (1,4)  (1,5)  (1,8)  (1,10)
(2,2)  (2,5)  (2,6)  (2,9)
(3,1)  (3,3)  (3,5)  (3,7)  (3,8)  (3,10)
(4,4)  (4,5)  (4,8)
(5,1)  (5,4)  (5,6)  (5,8)  (5,10)
(6,1)  (6,2)  (6,4)  (6,5)  (6,7)  (6,9)  (6,10)
(7,3)  (7,4)  (7,5)  (7,8)  (7,10)
(8,1)  (8,3)  (8,6);
```

Assegnamento di costo minimo di operai ad attività - descrizione

Il dirigente di un'azienda ha 8 attività da svolgere la prossima settimana e 8 operai a disposizione per svolgerle. Ogni operaio è in grado di svolgere tutte le attività ed il costo per far svolgere una attività ad un operaio è indicato in tabella:

	Attività							
Operai	1	2	3	4	5	6	7	8
1	14	14	11	14	10	15	13	10
2	13	7	9	12	15	8	11	7
3	8	10	9	15	9	9	13	15
4	15	7	8	15	8	10	13	13
5	12	7	15	7	13	12	9	11
6	9	12	11	9	15	13	10	8
7	12	15	11	9	12	9	7	15
8	9	10	9	10	15	14	15	13

Ogni operaio deve svolgere una sola attività ed ogni attività deve essere svolta da un solo operaio.
Il dirigente deve decidere quale attività assegnare ad ogni operaio in modo da svolgere tutte le attività con l'obiettivo di minimizzare il costo totale.

Assegnamento di costo minimo di operai ad attività - formulazione

Se O è l'insieme degli operai e T l'insieme delle attività, allora possiamo definire un grafo bipartito (N, A), dove
l'insieme dei nodi $N = O \cup T$,
l'insieme degli archi $A = \{(i,j) : \ i \in O, \ j \in T\}$



Assegnamento di costo minimo di operai ad attività - formulazione

Parametri: c_{ij} = costo per far svolgere l'attività j all'operaio i

Variabili decisionali: $x_{ij} = \begin{cases} 1 & \text{se l'operaio } i \text{ svolge l'attività } j \\ 0 & \text{altrimenti} \end{cases}$

per ogni arco $(i, j) \in A$.

Funzione obiettivo: $\sum_{(i,j)\in A} c_{ij}x_{ij}$ (da minimizzare)

Vincoli:

ogni operaio deve svolgere una sola attività:

$x_{1,1} + x_{1,2} + x_{1,3} + x_{1,4} + x_{1,5} + x_{1,6} + x_{1,7} + x_{1,8} = 1$

$x_{2,1} + x_{2,2} + x_{2,3} + x_{2,4} + x_{2,5} + x_{2,6} + x_{2,7} + x_{2,8} = 1$

...

ogni attività deve essere svolta da un solo operaio:

$x_{1,1} + x_{2,1} + x_{3,1} + x_{4,1} + x_{5,1} + x_{6,1} + x_{7,1} + x_{8,1} = 1$

$x_{1,2} + x_{2,2} + x_{3,2} + x_{4,2} + x_{5,2} + x_{6,2} + x_{7,2} + x_{8,2} = 1$

...

Assegnamento di costo minimo di operai ad attività - formulazione

Modello matematico (assegnamento di costo minimo)

$$
\begin{cases}
\min \displaystyle\sum_{(i,j)\in A} c_{ij} x_{ij} \\
\displaystyle\sum_{j\in T} x_{ij} = 1 \qquad \forall\ i \in O \\
\displaystyle\sum_{i\in O} x_{ij} = 1 \qquad \forall\ j \in T \\
x_{ij} \in \{0,1\} \qquad \forall\ (i,j) \in A
\end{cases}
$$

Assegnamento di costo minimo di operai ad attività - formulazione AMPL

assegn2.mod

```
#----------- parametri --------------------

set O; # insieme degli operai
set T;  # insieme delle attività
param c{i in O, j in T};

#----------- variabili --------------------

var x{i in O, j in T} binary;

#----------- funzione obiettivo --------------------

minimize Costo: sum{i in O, j in T} c[i,j]*x[i,j];

#----------- vincoli ------------------

s.t. v_operai{i in O}: sum{j in T} x[i,j] = 1;

s.t. v_attivita{j in T}: sum{i in O} x[i,j] = 1;
```

Assegnamento di operai ad attività - formulazione AMPL

assegn2.dat

```
   set O := 1 2 3 4 5 6 7 8;

   set T := 1 2 3 4 5 6 7 8;

param c : 1 2 3 4 5 6 7 8 :=
1 14 14 11 14 10 15 13 10
2 13 7 9 12 15 8 11 7
3 8 10 9 15 9 9 13 15
4 15 7 8 15 8 10 13 13
5 12 7 15 7 13 12 9 11
6 9 12 11 9 15 13 10 8
7 12 15 11 9 12 9 7 15
8 9 10 9 10 15 14 15 13;
```

Problema di trasporto - descrizione

Un'azienda automobilistica deve consegnare in 5 diverse città le automobili che ha prodotto in 2 stabilimenti. Il grafo seguente mostra gli stabilimenti (nodi 1 e 7), le città di destinazione (nodi 2, 3, 4, 5, 6) ed i loro collegamenti possibili.
Il numero (rosso) associato ad ogni nodo rappresenta l'offerta (valore negativo) o la domanda (valore positivo) del nodo.
Il numero (verde) associato ad ogni arco rappresenta il costo per spedire un'auto lungo quell'arco.



L'azienda deve decidere come spedire le auto dai 2 stabilimenti in modo da rispettare l'offerta e la domanda di tutti i nodi, con l'obiettivo di minimizzare il costo totale di trasporto.

Problema di trasporto - formulazione

Parametri: c_{ij} = costo per spedire un'auto da i a j

Variabili decisionali: x_{ij} = numero di automobili spedite lungo l'arco (i, j), per ogni arco $(i, j) \in A$.

Funzione obiettivo: $\sum_{(i,j)\in A} c_{ij} x_{ij}$ (da minimizzare)

Vincoli:

$x_{1,2} + x_{1,4} \leq 200$ (nodo 1)

$x_{1,2} - x_{2,3} \geq 100$ (nodo 2)

$x_{2,3} + x_{5,3} - x_{3,5} \geq 60$ (nodo 3)

$x_{1,4} + x_{5,4} + x_{7,4} \geq 80$ (nodo 4)

$x_{3,5} + x_{6,5} + x_{7,5} - x_{5,3} - x_{5,4} - x_{5,6} \geq 170$ (nodo 5)

$x_{5,6} + x_{7,6} - x_{6,5} \geq 70$ (nodo 6)

$x_{7,4} + x_{7,5} + x_{7,6} \leq 300$ (nodo 7)

Problema di trasporto - formulazione

Modello matematico

$$
\begin{cases}
\min \displaystyle\sum_{(i,j)\in A} c_{ij} x_{ij} \\
x_{1,2} + x_{1,4} \leq 200 \\
x_{1,2} - x_{2,3} \geq 100 \\
x_{2,3} + x_{5,3} - x_{3,5} \geq 60 \\
x_{1,4} + x_{5,4} + x_{7,4} \geq 80 \\
x_{3,5} + x_{6,5} + x_{7,5} - x_{5,3} - x_{5,4} - x_{5,6} \geq 170 \\
x_{5,6} + x_{7,6} - x_{6,5} \geq 70 \\
x_{7,4} + x_{7,5} + x_{7,6} \leq 300 \\
x_{ij} \in \mathbb{Z} \qquad \forall\ (i,j) \in A \\
x_{ij} \geq 0 \qquad \forall\ (i,j) \in A
\end{cases}
$$

Problema di trasporto - formulazione AMPL

trasp.mod

```
#----------- parametri -------------------

param n integer > 0; # numero di nodi
set N := 1..n; # insieme dei nodi
set A within (N cross N); # insieme degli archi
param c{(i,j) in A}; # c[i,j] = costo unitario dell'arco (i,j)
param bilancio{i in N}; # bilancio[i] = bilancio del nodo i

#------------ variabili -------------------

var x{(i,j) in A} integer >= 0;

#------------ funzione obiettivo -------------------

minimize CostoTotale: sum{(i,j) in A} c[i,j]*x[i,j] ;

#------------ vincoli -------------------

s.t. v_bilancio{i in N}:
sum{(j,i) in A} x[j,i] - sum{(i,j) in A} x[i,j] >= bilancio[i];
```

Problema di trasporto - formulazione AMPL

trasp.dat

```
param n := 7;
set A := (1,2) (1,4) (2,3) (3,5) (5,3) (5,4) (5,6) (6,5) (7,4) (7,5) (7,6);
param c :=
1 2 30
1 4 40
2 3 50
3 5 35
5 3 40
5 4 30
5 6 35
6 5 25
7 4 50
7 5 45
7 6 50;
param bilancio :=
1 -200
2 100
3 60
4 80
5 170
6 70
7 -300;
```

Problema di cammino minimo - descrizione

Supponiamo che il grafo seguente rappresenti una rete stradale che collega tra loro 11 città (rappresentate dai nodi).



Il numero associato ad ogni arco (i, j) rappresenta il tempo di viaggio (in ore) necessario per andare da i a j.
Vogliamo trovare il percorso più veloce che colleghi la città 1 con la città 11.

Problema di cammino minimo - formulazione

In un grafo orientato (N, A), un cammino orientato è una sequenza di nodi $i_1, i_2, \ldots, i_p \in N$ tale che $(i_1, i_2) \in A$, $(i_2, i_3) \in A$, $\ldots$, $(i_{p-1}, i_p) \in A$.
Pertanto dobbiamo trovare un cammino orientato dal nodo 1 al nodo 11.

Variabili decisionali: $x_{ij} = \begin{cases} 1 & \text{se l'arco } (i,j) \text{ appartiene al cammino} \\ 0 & \text{altrimenti} \end{cases}$

Funzione obiettivo:

$2.5x_{1,2} + 3x_{1,3} + 1.7x_{2,3} + 2.5x_{2,4} + 1.7x_{3,5} + 2.8x_{3,6} + 2x_{4,6} + 1.5x_{4,7} + 2x_{5,6} + 5x_{5,9} + 3x_{6,8} + 4.7x_{6,9} + 1.5x_{7,8} + 2.3x_{7,10} + 2x_{8,9} + 1.1x_{8,10} + 3.3x_{9,11} + 2.7x_{10,11}$ (da minimizzare)

Vincoli:

$$\begin{aligned}
& -x_{1,2} - x_{1,3} = -1 && \text{(nodo 1)} \\
& x_{1,2} - x_{2,3} - x_{2,4} = 0 && \text{(nodo 2)} \\
& x_{1,3} + x_{2,3} - x_{3,5} - x_{3,6} = 0 && \text{(nodo 3)} \\
& x_{2,4} - x_{4,6} - x_{4,7} = 0 && \text{(nodo 4)} \\
& x_{3,5} - x_{5,6} - x_{5,9} = 0 && \text{(nodo 5)} \\
& x_{3,6} + x_{4,6} + x_{5,6} - x_{6,8} - x_{6,9} = 0 && \text{(nodo 6)} \\
& x_{4,7} - x_{7,8} - x_{7,10} = 0 && \text{(nodo 7)} \\
& x_{6,8} + x_{7,8} - x_{8,9} - x_{8,10} = 0 && \text{(nodo 8)} \\
& x_{5,9} + x_{6,9} + x_{8,9} - x_{9,11} = 0 && \text{(nodo 9)} \\
& x_{7,10} + x_{8,10} - x_{10,11} = 0 && \text{(nodo 10)} \\
& x_{9,11} + x_{10,11} = 1 && \text{(nodo 11)}
\end{aligned}$$

Problema di cammino minimo - formulazione

Modello matematico (cammino di costo minimo)

$$
\begin{cases}
\min \displaystyle\sum_{(i,j)\in A} c_{ij}x_{ij} \\
\displaystyle\sum_{j:\ (j,1)\in A} x_{j1} - \sum_{j:\ (1,j)\in A} x_{1j} = -1 \\
\displaystyle\sum_{j:\ (j,i)\in A} x_{ji} - \sum_{j:\ (i,j)\in A} x_{ij} = 0 & \forall\ i \neq 1, 11 \\
\displaystyle\sum_{j:\ (j,11)\in A} x_{j,11} - \sum_{j:\ (11,j)\in A} x_{11,j} = 1 \\
x_{ij} \in \{0,1\} \qquad \forall\ (i,j) \in A
\end{cases}
$$

Problema di cammino minimo - formulazione AMPL

cm.mod

```
#---------- parametri -------------------
param n integer > 0; # numero di nodi
set N := 1..n; # insieme dei nodi
set A within (N cross N); # insieme degli archi
param costo{(i,j) in A}; # costo[i,j] = costo dell'arco (i,j)
param origine in N; # origine del cammino
param destinazione in N; # destinazione del cammino
#----------- variabili -------------------
var x{(i,j) in A} binary;
#----------- funzione obiettivo -------------------
minimize CostoTotale: sum{(i,j) in A} costo[i,j]*x[i,j];
#----------- vincoli -------------------
s.t. bilancio_origine:
sum{(j,origine) in A} x[j,origine] - sum{(origine,j) in A} x[origine,j] = -1;

s.t. bilancio_intermedio{i in N: i<>origine and i<>destinazione}:
sum{(j,i) in A} x[j,i] - sum{(i,j) in A} x[i,j] = 0;

s.t. bilancio_destinazione:
sum{(j,destinazione) in A} x[j,destinazione]
- sum{(destinazione,j) in A} x[destinazione,j] = 1;
```

Problema di cammino minimo - formulazione AMPL

cm.dat

```
param n := 11;
set A := (1,2) (1,3) (2,3) (2,4) (3,5) (3,6) (4,6) (4,7)
(5,6) (5,9) (6,8) (6,9) (7,8) (7,10) (8,9) (8,10) (9,11) (10,11);
param costo :=
1 2 2.5
1 3 3
2 3 1.7
2 4 2.5
3 5 1.7
3 6 2.8
4 6 2
4 7 1.5
5 6 2
5 9 5
6 8 3
6 9 4.7
7 8 1.5
7 10 2.3
8 9 2
8 10 1.1
9 11 3.3
10 11 2.7;
param origine := 1; param destinazione := 11;
```

Esercizio

In un ufficio devono essere pianificati il rimpiazzo e la manutenzione della fotocopiatrice per i prossimi quattro anni. All'inizio del primo anno viene sostituita la precedente fotocopiatrice con una nuova. I costi di manutenzione aumentano al passare degli anni di utilizzo della stessa fotocopiatrice, in quanto la fotocopiatrice diventa sempre meno affidabile. I costi di manutenzione sono indicati in tabella:

Manutenzione	Costo
1° anno	500 €
2° anno	1000 €
3° anno	1500 €
4° anno	2100 €

Ogni anno esiste la possibilità di sostituire l'attuale fotocopiatrice con una completamente nuova pagando sempre 1600 €. Ad esempio, se si rimpiazza la fotocopiatrice ogni anno il costo è di (1600+500) € per 4 anni; invece se viene rimpiazzata solo all'inizio il costo è (1600+500+1000+1500+2100) €.
Il capo dell'ufficio deve decidere quante volte sostituire la fotocopiatrice e in quale anno, in modo da minimizzare il costo totale.
Formulare tale problema come un problema di cammino minimo su un opportuno grafo e trovare la soluzione ottima.

Pianificazione di un progetto - descrizione

Il dirigente di un'azienda deve gestire un progetto costituito da 6 attività (A, B, C, D, E, F). È noto il tempo necessario per svolgere ogni attività (tempo di processamento). Le 6 attività non possono iniziare contemporaneamente perché sono soggette a vincoli di precedenza, ad esempio l'attività E può iniziare solo dopo che è stata completata l'attività D. I tempi di processamento delle attività e gli eventuali predecessori di ogni attività sono indicati in tabella:

attività	A	B	C	D	E	F
tempo di processamento	14	5	3	7	4	10
predecessori	–	–	A,B	A	D	C,E

Il dirigente deve decidere quando iniziare ogni attività, rispettando i vincoli di precedenza, in modo da completare tutte le attività del progetto nel più breve tempo possibile.

Pianificazione di un progetto - formulazione

Definiamo il grafo (N, A), dove:

insieme dei nodi $N = \{$origine o, destinazione d, attività $j\}$

insieme degli archi $A = \left\{ \begin{array}{ll} (o,j), & \text{se attività } j \text{ non ha predecessori,} \\ (i,j), & \text{se attività } j \text{ deve essere eseguita dopo attività } i, \\ (i,d), & \text{se attività } i \text{ non ha successori} \end{array} \right\}$

costi degli archi: $c_{oj} = 0$, $c_{ij} = p_i$ (tempo di processamento di i), $c_{id} = p_i$

Pianificazione di un progetto - formulazione

Definiamo il grafo (N, A), dove:
insieme dei nodi $N = \{\text{origine } o, \text{ destinazione } d, \text{ attività } j\}$

$$\text{insieme degli archi } A = \left\{ \begin{array}{ll} (o,j), & \text{se attività } j \text{ non ha predecessori,} \\ (i,j), & \text{se attività } j \text{ deve essere eseguita dopo attività } i, \\ (i,d), & \text{se attività } i \text{ non ha successori} \end{array} \right\}$$

costi degli archi: $c_{oj} = 0$, $c_{ij} = p_i$ (tempo di processamento di i), $c_{id} = p_i$



Pianificazione di un progetto - formulazione

Definiamo il grafo (N, A), dove:

insieme dei nodi $N = \{$origine o, destinazione d, attività $j\}$

$$\text{insieme degli archi } A = \left\{ \begin{array}{ll} (o,j), & \text{se attività } j \text{ non ha predecessori,} \\ (i,j), & \text{se attività } j \text{ deve essere eseguita dopo attività } i, \\ (i,d), & \text{se attività } i \text{ non ha successori} \end{array} \right\}$$

costi degli archi: $c_{oj} = 0$, $c_{ij} = p_i$ (tempo di processamento di i), $c_{id} = p_i$

Il problema di completare il progetto nel più breve tempo possibile equivale a cercare sul grafo (N, A) un cammino di costo massimo da o a d.



Pianificazione di un progetto - formulazione

Ogni cammino di costo massimo è detto cammino critico perché un ritardo sull'esecuzione dei nodi (attività) che lo formano causa un ritardo sul completamento dell'intero progetto.



cammino critico: o-A-D-E-F-d, costo $35 =$ minima durata del progetto.

Pianificazione di un progetto - formulazione

Quando devo iniziare ad eseguire un'attività non critica in modo da non causare un ritardo all'intero progetto?

min istante inizio $j =$
max istante inizio $j =$



Pianificazione di un progetto - formulazione

Quando devo iniziare ad eseguire un'attività non critica in modo da non causare un ritardo all'intero progetto?

min istante inizio $j =$ costo cammino max da o a j.
max istante inizio $j =$



Pianificazione di un progetto - formulazione

Quando devo iniziare ad eseguire un'attività non critica in modo da non causare un ritardo all'intero progetto?

min istante inizio j = costo cammino max da o a j.
max istante inizio j = min durata progetto − costo cammino max da j a d.



Pianificazione di un progetto - formulazione

Quando devo iniziare ad eseguire un'attività non critica in modo da non causare un ritardo all'intero progetto?

min istante inizio j = costo cammino max da o a j.
max istante inizio j = min durata progetto − costo cammino max da j a d.



Attività B: min ist. inizio = 0, max ist. inizio = 35 − 18 = 17.
Attività C: min ist. inizio = 14, max ist. inizio = 35 − 13 = 22.
Se l'attività B inizia all'istante 17, allora l'attività C non può essere ritardata.

Pianificazione di un progetto - risoluzione

Osserviamo che a causa dei vincoli di precedenza tra le attività, il grafo (N, A) non può contenere cicli (cioè cammini chiusi).

Quindi nel grafo esiste un ordine topologico, cioè ad ogni nodo può essere associato un numero intero in modo che

$$\text{se } (i, j) \in A, \text{ allora } i < j.$$

Metodo di ordinamento topologico

1. Poni $k := 1$
2. Assegna k ad un nodo i che non ha predecessori
3. Elimina dal grafo il nodo i e tutti gli archi uscenti da i
4. Se il grafo è vuoto allora STOP
   altrimenti poni $k := k + 1$ e torna al passo 2.

Pianificazione di un progetto - risoluzione

Esempio



Pianificazione di un progetto - risoluzione

Esempio



Pianificazione di un progetto - risoluzione

Esempio



Pianificazione di un progetto - risoluzione

Esempio



Pianificazione di un progetto - risoluzione

Esempio



Pianificazione di un progetto - risoluzione

Esempio



Pianificazione di un progetto - risoluzione

Esempio



Pianificazione di un progetto - risoluzione

Esempio



Pianificazione di un progetto - risoluzione

Esempio



Pianificazione di un progetto - risoluzione

Esempio



Pianificazione di un progetto - risoluzione

Esempio



Pianificazione di un progetto - risoluzione

Esempio



Pianificazione di un progetto - risoluzione

Esempio



Pianificazione di un progetto - risoluzione

Esempio



Pianificazione di un progetto - risoluzione

Esempio



Pianificazione di un progetto - risoluzione

Esempio



Pianificazione di un progetto - risoluzione

Con un algoritmo di programmazione dinamica è possibile trovare:

- la minima durata del progetto
- il minimo istante inizio a_j ed il massimo istante inizio b_j di ogni attività j

Metodo CPM (Critical Path Method)

1. Ordina topologicamente i nodi ($1 =$ origine, $n =$ destinazione)
2. (Calcola il minimo istante di inizio di ogni attività in avanti)

   $a_1 = 0$

   for $j = 2, \dots, n$ **do**

   $$a_j = \max_{i:\ (i,j) \in A} \{a_i + c_{ij}\}$$

   end
3. (Calcola il massimo istante di inizio di ogni attività all'indietro)

   $b_n = a_n$

   for $j = n-1, \dots, 1$ **do**

   $$b_j = \min_{k:\ (j,k) \in A} \{b_k - c_{jk}\}$$

   end

Pianificazione di un progetto - risoluzione

Esempio

Applichiamo il metodo CPM per trovare il minimo e massimo istante di inizio di tutte le attività del seguente progetto:

attività	A	B	C	D	E	F
tempo di processamento	14	5	3	7	4	10
predecessori	–	–	A,B	A	D	C,E



Pianificazione di un progetto - risoluzione

Esempio (segue)

Ordino topologicamente i nodi:



j	a_j	b_j
1	0	$\min\{0, 17\} = 0$
2	0	$\min\{0, 8\} = 0$
3	0	17
4	14	14
5	$\max\{14, 5\} = 14$	22
6	21	21
7	$\max\{25, 17\} = 25$	25
8	35	35

Pianificazione di un progetto

Esercizio.
Consideriamo il problema di pianificare un progetto in cui le attività da eseguire sono elencate in tabella:

attività	A	B	C	D	E	F	G	H	I
tempo di processamento	120	60	15	45	30	60	40	30	40
predecessori	–	–	B	B	F	–	A	E,I,G	B,D,G

1. Applicare il metodo CPM per trovare il minimo e il massimo istante di inizio di ogni attività.
2. Qual è la minima durata del progetto?
3. Quali sono le attività critiche?